Mortgage payment

Suppose you take out a **30 year mortgage** for **$200,000** from a bank to buy a house.
Suppose the bank charges you interest of $\boldsymbol{r = 4\%}$ per year compounded daily.
Suppose you pay the mortgage **$P per year** in weekly payments (so $P/52 per week).

Let $M(t)$ be what you owe at time t years.
So $M(0) = \$200$ thousand (the mortgage is for $200,000)
And $M(30) = \$0$ thousand (it's a 30 year mortage)

The DE is

$$M'(t) = r \cdot M(t) - P$$

The initial value is $M(0) = \$200$ thousand.

1: input and solve the differential equation with $r = 0.04$ (interest payment)
2: solve the $M(30) = 0$ for P to determine the anual payment.
3: graph the solution $M(t)$ with this value for P
4: calculate the amount of interest paid $= 30 \cdot P - \$200000$
5: starting with $M = 200,000$ simulate the actual interest charges and actual weekly payments for 30 years

Solution

```
> restart;
  de := diff( M(t),t) = r*M(t) - P;
```

$$de := \frac{d}{dt} M(t) = r M(t) - P$$

```
> r := 0.04;
```

$$r := 0.04$$

1: Solve the differential equation for M(t) not for P yet

```
> sol := dsolve( {de,M(0)=200000}, M(t) );
```

$$sol := M(t) = 25 P + e^{\frac{1}{25} t} (200000 - 25 P)$$

2: Now M(30) = 0

```
> eval( sol, t=30 );
```

$$M(30) = 25 P + e^{\frac{6}{5}} (200000 - 25 P)$$

So we need to equate the right-hand-side of this equation to 0

```
> eqn := rhs( eval(sol, t=30) ) = 0;
```

$$eqn := 25 P + e^{\frac{6}{5}} (200000 - 25 P) = 0$$

```
> P := solve(eqn,P):
> P := evalf( P );
```

$$P := 11448.10209$$

```
> plot( rhs(sol), t=0..30 );
```



4: Total interest paid over 30 years (it's a lot, even at 4%)

```
> Interest := 30*P-200000;
```

$143,443.06

5: Simulation

```
> P := 11448.10;
```

$11,448.10

When I went to do the simulation, I thought that there were 52 weeks in a year.   But it's actually 52 weeks plus one day.

```
> 365/7.0;
```

52.14285714

```
> M := 200000;
  for d from 1 to 30*365 do
     M := M + (r/365)*M;
     if d mod 7 = 0 then M := M - P/365*7 fi;
  od:
  M;
```

$M := 200000$

299.9136943

That simulates a weekly payment after interest is charged on the 7th day. Here is what happens if we make the payment after three days, so in the middle of the week.

```
> M := 200000;
  for d from 1 to 30*365 do
     M := M + (r/365)*M;
     #if d mod 7 = 0 then M := M - P/365*7 fi;
     if d mod 7 = 3 then M := M - P/365*7 fi;
  od:
  M;
```

$M := 200000$

8.939097709